
1. Introducción y objetivos

El propósito general de esta sesión de laboratorio es operar con sistemas digitales modernos. Este 

propósito se concreta en los siguientes objetivos: 

▪ Conocer cómo opera un sistema digital con la estructura “Unidad de Datos y Unidad de Control”. 

En particular, el sistema propuesto realiza la multiplicación de dos números binarios A y B de 

cuatro bits mediante el algoritmo basado en sumas y desplazamientos a la derecha. 

▪ Desarrollar las habilidades de diseño e implementación. Para ello, tendrá que realizar la unidad de 

control del multiplicador secuencial, contando previamente con la especificación completa de la 

unidad de datos y con una descripción del microprograma de control. Además, se implementará el 

multiplicador para lo que se volcará el diseño sobre una FPGA. 

▪ Comprobar la implementación del multiplicador realizada sobre FPGA tanto a nivel  de micro-

operación como a nivel de macrooperación. 

Para la realización de este laboratorio se facilitan un conjunto de ficheros, algunos de ellos deben ser  

completados antes de la sesión de laboratorio. La tabla 1 resume el contenido y el objetivo de cada uno 

de ellos.

Nombre del 
fichero

Contenido Descripción

u_control.v Unidad de control del 
multiplicador

Debe completarlo el alumno antes de 
asistir a la sesión de laboratorio

u_datos.v Unidad de datos del multiplicador Debe utilizarlo sin modificaciones para 

Nombre del 
fichero

Contenido Descripción

conectarlo a la unidad de control.

multiplicador.v Módulo para el multiplicador
Debe completar la descripción estructural  
durante la sesión de laboratorio.

multiplicador_tb.v Testbench para multiplicador Debe utilizarlo para realizar la simulación

sistema_completo.v
Sistema completo para 
implementar en FPGA

Se utilizará sin modificaciones durante la 
implementación

basys2.ucf
Fichero con la descripción de pads 
de la placa Basys2

Se utilizará sin modificaciones durante la 
implementación

Tabla 1. Ficheros necesarios durante la sesión de laboratorio.

2. Estudio teórico

Estudie adecuadamente la descripción del multiplicador secuencial que aparece en la sección  3 y 

realice dos tareas:

1. A partir de la carta ASM propuesta para el multiplicador secuencial (figura 5), realice la unidad de 

control  en  Verilog  mediante  una  descripción  canónica.  Para  ello  utilice  el  código  del  fichero 

u_control.v  (ver código  1 mostrado al final de esta  sección). Este apartado es  absolutamente 

imprescindible para realizar la práctica. 

2. Usando las tablas proporcionadas en las últimas páginas,  muestre la secuencia de datos y de 

señales de control  del  sistema en cada ciclo de reloj  CLK, para las dos siguientes multiplica-

ciones :

2.1. A = 1011 B = 0001

2.2. A = 0010 B = 1101

module u_control(
 input xs,clk,reset,busc0,cycont,
 output reg clsincr,fin,
 output reg wa,wd,wsuml,wsumh,shrsum,upcont
    );

 // Defina aqui la lista de estados con la sentencia parameter
  
 ...

 // Defina aqui dos variables  tipo reg llamadas: 
 //     estado_actual y siguiente_estado
 // Dimensione ambas variables con el tamaño correcto en función
 // del número de estados que definió previamente
 
 ...

 // Proceso de cambio de estado, utiliza la variables
 // definidas previamente
 always @(posedge clk,posedge reset)
  begin
   if(reset)
    estado_actual <= S0;

  else
   estado_actual <= siguiente_estado;
  end 
  // Proceso combinacional de calculo de proximo estado
  // debe obtenerlos a partir de la carta ASM 
  always @(*)
  begin
   clsincr=0; // Por defecto se establecen todas las señales a cero
   wa=0;
   wd=0;
   wsuml=0;
   wsumh=0;
   upcont=0;
   shrsum=0;
   fin=0;
   case(estado_actual)
      // Rellene todos los estados y active las señales de control siguiendo la carta ASM
      S0:
        if(xs)
          begin
          clsincr=1;
          siguiente_estado = S1;
          end
        else
          siguiente_estado = S0;

       // Aqui deben ir el resto de estados
       ....
       ///

      default:  // No elimine esta sentencia para evitar problemas
          siguiente_estado=S0;  // Cuidado aqui, hay que cubrir todos los casos 
                                // de lo contrario aparece un latch
   endcase
  end
endmodule

Código 1. Fichero u_control.v, plantilla de código para la unidad de control.

3. Descripción del multiplicador secuencial

El multiplicador secuencial es un sistema digital con dos bloques bien diferenciados: Unidad de datos 

y Unidad de control (ver figura 1).

El  funcionamiento  global  de  este  multiplicador  es  el  siguiente.  La  primera  vez  que  se  dispone  a 

multiplicar,  el  usuario realizará un RESET del sistema activando la señal  destinada a ello (RESET).  

Posteriormente, activará la señal XS y la unidad de control irá proporcionando, ciclo a ciclo, las señales  

de control que necesita cada elemento de la unidad de datos para que se realice la multiplicación de los  

números A y B que se hayan fijado como entrada al sistema.

El circuito de la unidad de datos se muestra en la figura 2. Éste surge del algoritmo de multiplicación 

que se describe en la sección 3.1 Puede ver que está compuesto por tres registros (A, SUMH y SUML), 

un sumador paralelo de 4 bits y un contador módulo 4 (CONT). Cada uno de ellos tiene su tabla de  

descripción RT representada en la misma figura 3. Las salidas de estos registros son incondicionales.

Figura 1. Representación a nivel de bloques del multiplicador secuencial.

Figura 2. Representación estructural de la ruta de datos.

clk cl w Operación

1 - A ← 0

0 1 A ← DATA[3:0]

0 0 A ← A

Registro A

clk cl w shr Operación

1 - - R ← 0

0 1 - R ← Din[3:0]

0 0 1 R ← SHR(R,sri)

0 0 0 R ← R

Registro de desplazamiento

clk cl up Operación

1 - CONT ← 0

0 1 CONT ← CONT+1

0 0 CONT ← CONT

Contador

clk cl w Operación

1 - D ← 0

0 1 D ← Din

0 0 D ← D

Biestable

Figura 3. Descripción RT de los componentes de la ruta de datos.

3.1. Algoritmo de multiplicación

Para realizar la multiplicación de dos números binarios de 4 bits, A y B, se ha seguido el algoritmo de 

sumas y desplazamientos a la derecha.  A continuación se describe el procedimiento.

Cuando se realiza la multiplicación de dos números binarios mediante el procedimiento tradicional de 

multiplicaciones se puede observar como la multiplicación consiste en realizar sumas desplazadas a la  

izquierda.  Al  ser  los  datos  binarios,  los  sumandos  sólo  pueden  ser  el  primer  operando  de  la 

multiplicación o cero.

El algoritmo de multiplicación por sumas y desplazamientos propuesto consiste en realizar cada una de 

estas sumas parcialmente. La idea básica es sumar a un resultado parcial un operando cada vez que un 

bit del  otro operando sea 1.  En cada paso se realiza una suma parcial  y se guarda en un registro  

separado el bit menos significativo del resultado obtenido. Esto se debe a que en la siguiente suma 

parcial este bit ya no interviene. Realizando sucesivamente la operación suma y desplazamiento a la 

derecha de un bit se obtiene el resultado de la multiplicación.

En la figura 4 se muestra un ejemplo de multiplicación siguiendo dicho algoritmo paso a paso. Observe 

como en cada paso se analiza un bit del operando B. Siempre que encuentre un 1 realiza una suma y un 

desplazamiento de un bit (2 pasos). En caso se ser este bit  0, se realiza únicamente el desplazamiento 

de un bit (un único paso). 

            1010
          ✕ 0101
          ──────
            1010
           0000
          1010
       + 0000
       ─────────
         0110010

Procedimiento de multiplicación tradicional

Algoritmo de sumas y desplazamientos

   1010
 ✕ 0101
   ────
   0000

Paso 1.Inicio

➪

   1010
 ✕ 0101
   ────
   0000
 + 1010
   ────
  01010

Paso 2.Suma

➪

     1010│
   ✕  010│1
     ────┼─
     0101│0
      

Paso 3.Desplazamiento

➪

       1010│
    ✕    01│01
    ───────┼──
       0010│10
       

Paso 4.Desplazamiento

➪

➪

      1010│
    ✕   01│01
      ────┼──
      0010│10
    + 1010│
      ────┤
     01100

Paso 5.Suma

➪

        1010│
   ✕       0│101
     ───────┼───
        0110│010
        

Paso 6.Desplazamiento

➪

        1010│
   ✕        │0101
     ───────┼───
        0011│0010
        

Paso 7.Desplazamiento

Figura 4. Algoritmo de multiplicación por suma y desplazamiento.

Para la ruta de datos propuesta el algoritmo se realizará de la siguiente forma: 

1. Se realiza una carga en paralelo de los datos A y B en los registros A y SUML respectivamente.

2. Se ha dispuesto de un contador módulo 4 para contar los 4 desplazamientos necesarios. Será el bit 

de acarreo de este contador el que indique cuándo finaliza la multiplicación. Debe iniciarse este 

contador a cero.

3. Se procede a analizar cuál es el bit menos significativo de B (registro SUML0). Si es 0 no se hace  

nada, en cambio, si es 1 se realiza una suma del dato de A con el dato presente en un tercer 

registro (SUMH) destinado a almacenar los productos parciales de la multiplicación. Esta suma se 

guarda en el mismo registro SUMH.

4. Tras el paso anterior se realiza una operación de desplazamiento a la derecha de los registros  

SUMH y SUML conjuntamente. 

4.1. SUMH desplaza su bit LSB a SUML así, en la próxima operación de suma este bit no se 

suma. 

4.2. Este bit desplazado pasa a convertirse en el siguiente bit del resultado y queda guardado en 

SUML.

4.3. Se decrementa el contador módulo 4.

5. Se realizarán iterativamente los pasos 3 y 4 cuatro veces, ya que el dato B posee 4 bits. Cuando se 

active el bit de acarreo del contador módulo 4 finaliza la multiplicación. 

6. El resultado final de la operación quedará guardado en los registros SUMH y SUML. 

Figura 5. Carta ASM de datos y de control del multiplicador secuencial.

Una carta ASM que da respuesta al algoritmo y Unidad de Datos anteriores es la que se muestra en la 

figura 5. De ella se obtiene directamente la Unidad de Control, la cual deberá ser obtenida por el alumno 

usando una descripción canónica en Verilog.

3.2. Sistema completo en la placa de desarrollo

Para probar el  diseño digital  realizado se utilizará una placa desarrollo del fabricante Digilentic1 

1 Empresa dedicada al diseño en tecnologías basadas en FPGA y microcontroladores. Web: http://www.digilentinc.com

llamada  Basys2 que incluye un FPGA de 500.000 puertas y  algunos componentes electrónicos para 

entrada y  salida básica.  Para poder  interactuar con el  sistema utilizando la  placa de desarrollo,  es 

necesario añadir al diseño un procedimiento para introducir los datos a multiplicar, poder visualizar los 

resultados y consultar el estado del sistema. Esta placa de desarrollo dispone de algunos elementos de 

entrada y salida como son: 8 conmutadores, 4 displays de 7 segmentos, 8 leds y 4 pulsadores entre 

otros.

Mediante varios módulos descritos en Verilog y ya preparados (fichero sistema_completo.v) se construye 

un sistema digital como el indicado de manera esquemática en la figura 6. Esta figura muestra el módulo 

multiplicador construido por el alumno, conectado a un módulo capaz de controlar el display 7 segmen-

tos. Este controlador consta de 4 entradas de 4 bits cada una de ellas (d0, d1 d2 y d3) correspondién-

dose cada una de las entradas con uno de los dígitos mostrados en el display. Además, se ha conectado 

el  reloj,  la  señal  xs y  la  señal  reset a  pulsadores  para  controlar  el  funcionamiento  del  sistema 

manualmente. También, la señal fin y el reloj se interconecta a los leds para poder visualizar el efecto de 

pulsación del reloj e indicar la finalización de la multiplicación. Por último, los datos A y B se pueden 

introducir en binario mediante conmutadores.

Figura 6. Esquema del sistema digital completo para el multiplicador.

Figura 7. Fotografía de la placa de desarrollo Basys2

La figura  7 muestra una fotografía de la placa de desarrollo. Para este diseño digital los 4 primeros 

conmutadores permiten introducir en binario el dato A. Simultáneamente el primer dígito del display 

mostrará en hexadecimal el número establecido según se cambien estos conmutadores. Con el dato B 

sucede lo mismo,  sólo que con los últimos cuatro conmutadores y  el  segundo dígito del display.  El  

resultado (formado por SUMH y SUML) se irá mostrando ciclo a ciclo de reloj en los dos últimos dígitos 

del display.

Para operar, primero se configuran los valores de A y B con los conmutadores, tras esto, es necesario 

generar la señal  xs de comienzo y el reloj utilizando los botones. Los botones señalados en la figura 

introducen  un  1  lógico  cuando  se  pulsan  y  están  conectados  a  las  señales  reset,  xs,  y clk.  El 

procedimiento consiste en pulsar en primer lugar el botón xs y tras esto pulsar repetidamente el reloj 

(clk). 

Si  el  sistema  está  operando  correctamente,  por  cada  ciclo  de  reloj  un  led  diferente  se  iluminará 

indicando que se ha detectado la pulsación y, los dos últimos dígitos irán mostrando el resultado parcial. 

Cuando  se  active  la  señal  de  fin  se  iluminarán  todos  los  leds  indicando  que  ha  terminado  la 

multiplicación. En este momento el resultado de la multiplicación estará en los dos últimos dígitos del 

display.

4. Estudio Experimental

En la sección anterior se ha descrito el multiplicador como Sistema Digital. Éste consta dos partes: 

Unidad  de  Datos  (fichero  u_datos.v)  y  Unidad  de  Control  (fichero  u_control.v).  La  primera  se 

proporcionan  al  alumno  ya  diseñada  y  la  segunda,  el  alumno  debe  traerla  hecha.  En  la  parte 

experimental de la práctica hay que realizar las siguientes tareas:

1. Completar la descripción estructural del multiplicador y simular el multiplicador para verificar el  

clk
xs

reset

Dato A Dato B

A B RES

Leds

Conmutador
de encendido

Modo
de

programación

correcto funcionamiento.

2. Realizar la implementación del sistema completo en FPGA y verificar su funcionamiento.

A continuación se detalla cada tarea.

4.1. Verificación del multiplicador

Para verificar el correcto funcionamiento del multiplicador se va a realizar una simulación de 10 

multiplicaciones mediante un testbench proporcionado en el fichero multiplicador_tb.v. Para conseguir 

este objetivo siga los siguientes pasos:

1. Cree un nuevo proyecto  vacío en el  entorno ISE siguiendo los mismos pasos  de la  sesión de 

laboratorio anterior.

2. Establezca el modo de simulación en el proyecto (figura 

8). Añada al proyecto los 4 ficheros mediante la opción 

de menú Add Source del menú Project. Los ficheros son 

los  indicados  a  continuación  teniendo  en  cuenta  los 

siguientes aspectos en cada uno de ellos:

2.1. Fichero u_datos.v: Debe añadirlo sin modificarlo y 

como Implementation.

2.2. Fichero  u_control.v:  Debe  haber  completado  el 

fichero  en  la  realización  de  estudio  teórico.  Añádalo 

como Implementation.

2.3. Fichero multiplicador.v: Se completará posteriormente. Debe añadirlo como Implementation.

2.4. Fichero  multiplicador_tb.v: Debe añadirlo como  Simulation. Tras añadir este fichero debe 

aparecer como raíz del árbol de ficheros del proyecto, corresponde al llamado test. 

3. El siguiente paso es editar el fichero multiplador.v abriéndolo desde el árbol de proyecto de ISE. 

Debe completar los fragmentos indicados en el mismo fichero, concretamente hay que realizar la 

interconexión de la  unidad de datos  y  de control  tal  y  como se  mostraba en la  figura  1.  Es 

recomendable  visualizar  el  contenido  del  fichero  u_datos.v  y  tomar  como  ejemplo  el  módulo 

u_datos donde se realiza el  conexionado de la unidad de datos. Debe completar en el  fichero 

multiplicador.v siguiendo los pasos indicados a continuación:

3.1. Realizar el  conexionado de la unidad de datos con los cables internos ya definidos:  wa, 

upcont,  wsuml,  wsumh,  cycont,  clsincr  y shrsum. Debe realizar la  conexión directa (sin usar 

cables internos) de las entradas  a, b, databus  y clk del módulo  multiplicador con la unidad de 

datos. Tenga en cuenta que  databus se refiere al bus de salida de la unidad de datos, llamada 

busc en la unidad de datos.

3.2. Realizar la instancia de la unidad de control del mismo modo que se ha instanciado la unidad 

de datos. Debe interconectar todas las entradas y salidas de la unidad de control, tanto a los 

cables internos como las entradas/salidas del módulo multiplicador. 

Figura 8. Modo simulación

Modo simulación

module multiplicador(
 input  [3:0] a,      // Entrada del dato A
 input  [3:0] b,      // Entrada del dato B
 input  clk,          // Reloj del sistema
 input  xs,           // Señal de comienzo
 input  reset,        // Reset del sistema
 output fin,          // Señal de fin de operacion
 output [7:0] databus // Bus de datos de resultado (parcial y final)
 );

// Conexiones internas para interconectar los dos módulos
wire wa,wd,upcont,wsuml,wsumh,cycont,clsincr,shrsum;

// Instancia de la unidad de datos debe completar las conexiones
u_datos  U_DATOS(
 .data(a) 

    ...

        // Complete aqui todas las conexiones que faltan
        // Para ello puede abrir el fichero incluido: "u_datos.v"
        // y mirar la definicion del modulo "u_datos" que sirve 
        // como ejemplo de interconexión de módulos.
  );

/* Aqui debe instanciar la unidad de control del mimso
   modo que se ha instanciado la unidad de datos.
   
   Debe interconectar todas las conexiones 
   con la unidad de datos: tanto las entradas como las salidas. */

     ...

endmodule

Código 2. Fichero multiplicador.v, plantilla de código interconexión de unidad de datos y de control.

4. Por último se va a realizar la simulación. Seleccione el testbench en el árbol de proyecto, es el  

módulo superior en el árbol de proyecto (ver figura 9), con el nombre test (multiplicador_tb.v). 

4.1. Tras esta selección debe utilizar la entrada  Behavioural Check Syntax  para comprobar si 

hay errores en su código. Debe corregir todos los errores (ver indicación en la figura 9).

4.2. Una  vez  corregidos  los  errores,  realice  la  simulación  mediante  la  opción  Simulate 

Behavioural Model. Debe comprobar todas la señales que obtuvo en la tabla del estudio teórico, 

para ello, añada a la simulación las señales internas de los módulos de la ruta de datos tal y como 

se muestra en el figura 10. Busque las multiplicaciones del estudio teórico y amplíe la forma de 

onda hasta que se visualicen correctamente las señales.

Figura 9. Simulación del tesbench. Figura 10. Acceso a módulos internos.

4.2. Implementación del multiplicador en FPGA

Finalmente se implementará el sistema digital realizado en un dispositivo programable tipo FPGA 

incluida en la placa de desarrollo Basys2.

Tal y como se detalló en la sección  3.2 el multiplicador necesita conec-

tarse a los conmutadores, botones y el display de la placa de desarrollo 

para  poder  interactuar  con  él.  Los  módulos  que  controlan  estos 

componentes  se  encuentran  ya  diseñados  y  testados  en  el  fichero 

sistema_completo.v. Además del sistema completo,  es necesario utilizar 

otro fichero donde se indican las conexiones entre los pads del chip FPGA 

y  los  componentes  de  la  placa.  Este  fichero  se  llama  basys2.ucf y  ya 

contiene la asignación de las conexiones del sistema a los componentes de 

la placa.

Los pasos a seguir son los siguientes:

1. Cambiar el proyecto ISE del modo simulación al modo implemen-

tación tal y como se mostraba en la figura 8

2. Añadir dos ficheros al proyecto:

2.1. Fichero  sistema_completo.v como  implementación.  Este 

fichero aparecerá ahora como raíz del proyecto.

2.2. Fichero basys2.ucf.  Debe comprobar que también aparece en 

el árbol de proyecto.

Selección de 
componentes internos

Figura 11. Implementación

3. La implementación se realiza seleccionado el módulo  sistema_completo en el árbol de proyecto. 

Debe seleccionar la opción Generate Programming File (figura 11) y seguir los siguientes pasos:

3.1. Pulsando el botón derecho del ratón sobre la opción Generate Programming File aparecerá 

un menú flotante como el mostrado en la figura 12a, seleccionando la opción de menú de Process 

Poperties aparecerá el diálogo mostrado en la figura 12b.

3.2. En el  diálogo hay que elegir la categoría  Startup Options y  cambiar el  valor del primer 

campo FPGA Start-Up Clock de CCLK a Jtag-Clock. Tras aceptar los cambios con el botón OK se 

volverá a la ventana principal de ISE. 

3.3. Pulsando dos veces botón izquierdo del ratón sobre Generate Programming File se ejecuta el 

proceso completo y se genera el fichero de programación. Tal y como se muestra en la figura 11, 

si el proceso ha terminado con éxito aparecerá un indicador verde, en caso contrario un aspa rojo 

indicando  la  existencia  de  errores.  En  caso  de  existir  errores  debe  corregirlos  y  repetir  el 

proceso.

(a) (b)

Figura 12. Opciones de generación del fichero de programación: 
(a) menú desplegable, (b) diálogo con opciones

4. El último paso consiste en programar la placa de desarrollo con un fichero que se ha generado 

tras el proceso del paso anterior. Concretamente, el fichero debería llamarse sistema_completo.bit 

y hay que transferirlo por la conexión USB a la FPGA. Para ello siga los siguientes pasos:

4.1. Compruebe en la placa el modo de programación establecido, para ello fíjese en la figura 7 

donde está situado el conmutador Modo de Programación. Asegúrese que está en modo PC.

4.2. Conecte el puerto USB de la placa de desarrollo y conmute la alimentación de la placa. Por 

defecto debe cargarse un programa de test  de la propia placa que cuenta dígitos BCD en el 

Display. Además, puede comprobar el correcto funcionamiento de los conmutadores y los botones 

antes de proceder con la programación.

4.3. Inicie el programa  Adept desde el menú de inicio (menú  Digilent →  Adept, icono  ) y 

aparecerá el programa mostrado en la figura 13. Con este programa se puede transferir el fichero 

de programación (bitstream) a la FPGA. El programa Adept permite programar los componentes 

de la placa Basys2, estos son, una FPGA y una PROM. Se programará la FPGA, por tanto, se debe 

utilizar el botón Browse indicado en la figura y seleccionar el fichero .bit. Hay buscar la carpeta 

del proyecto ISE en el que está trabajando, allí encontrará el resultado de la síntesis en un fichero 

Opciones de
programación

.bit, concretamente, sistema_completo.bit. Una vez seleccionado este fichero se activará el botón 

Program y bastará con pulsarlo para la placa se programe.

Figura 13. Programación de placas Digilent con Adept.

5. En la figura 7 se mostraba la interconexión de los componentes de la placa de desarrollo con el 

módulo  multiplicador  desarrollado.  Fijándose  en  esta  figura,  utilice  los  conmutadores  y  los 

botones para realizar la multiplicación de varios números y compruebe si el resultado es correcto. 

El procedimiento es:

5.1. Colocar los números con los conmutadores.

5.2. Pulsar el botón xs para que comience la multiplicación.

5.3. Generar el reloj manualmente pulsando el último botón repetidamente hasta que se iluminen 

todos los Leds, lo que indica que se ha activado la señal fin y ha terminado la multiplicación.

Botón para 
programación

Selección de
fichero

# This file is a general .ucf for Basys2 rev C board

# To use it in a project:

# - remove or comment the lines corresponding to unused pins

# - rename the used signals according to the project



# clock pin for Basys2 Board

NET "clk" LOC = "B8"; # Bank = 0, Signal name = MCLK

NET "clk" CLOCK_DEDICATED_ROUTE = FALSE;



# Pin assignment for DispCtl

# Connected to Basys2 onBoard 7seg display

NET "seg<0>" LOC = "L14"; # Bank = 1, Signal name = CA

NET "seg<1>" LOC = "H12"; # Bank = 1, Signal name = CB

NET "seg<2>" LOC = "N14"; # Bank = 1, Signal name = CC

NET "seg<3>" LOC = "N11"; # Bank = 2, Signal name = CD

NET "seg<4>" LOC = "P12"; # Bank = 2, Signal name = CE

NET "seg<5>" LOC = "L13"; # Bank = 1, Signal name = CF

NET "seg<6>" LOC = "M12"; # Bank = 1, Signal name = CG

NET "dp" LOC = "N13"; # Bank = 1, Signal name = DP



NET "an<3>" LOC = "K14"; # Bank = 1, Signal name = AN3

NET "an<2>" LOC = "M13"; # Bank = 1, Signal name = AN2

NET "an<1>" LOC = "J12"; # Bank = 1, Signal name = AN1

NET "an<0>" LOC = "F12"; # Bank = 1, Signal name = AN0



# Pin assignment for LEDs

NET "ld<7>" LOC = "G1" ; # Bank = 3, Signal name = LD7

NET "ld<6>" LOC = "P4" ; # Bank = 2, Signal name = LD6

NET "ld<5>" LOC = "N4" ;  # Bank = 2, Signal name = LD5

NET "ld<4>" LOC = "N5" ;  # Bank = 2, Signal name = LD4

NET "ld<3>" LOC = "P6" ; # Bank = 2, Signal name = LD3

NET "ld<2>" LOC = "P7" ; # Bank = 3, Signal name = LD2

NET "ld<1>" LOC = "M11" ; # Bank = 2, Signal name = LD1

NET "ld<0>" LOC = "M5" ;  # Bank = 2, Signal name = LD0



# Pin assignment for SWs

NET "data<3>" LOC = "N3";  # Bank = 2, Signal name = SW7

NET "data<2>" LOC = "E2";  # Bank = 3, Signal name = SW6

NET "data<1>" LOC = "F3";  # Bank = 3, Signal name = SW5

NET "data<0>" LOC = "G3";  # Bank = 3, Signal name = SW4

NET "datb<3>" LOC = "B4";  # Bank = 3, Signal name = SW3

NET "datb<2>" LOC = "K3";  # Bank = 3, Signal name = SW2

NET "datb<1>" LOC = "L3";  # Bank = 3, Signal name = SW1

NET "datb<0>" LOC = "P11";  # Bank = 2, Signal name = SW0



#NET "clk" LOC = "A7";  # Bank = 1, Signal name = BTN3

NET "reset" LOC = "M4";  # Bank = 0, Signal name = BTN2

NET "start" LOC = "C11"; # Bank = 2, Signal name = BTN1

NET "clkext" LOC = "G12"; # Bank = 0, Signal name = BTN0

NET "clkext" CLOCK_DEDICATED_ROUTE = FALSE;








// Design: multiplicador
// Description: Interconexión de la unidad de datos y de control
// Author: Paulino Ruiz de Clavijo Vázquez <paulino@dte.us.es>
// Copyright Universidad de Sevilla, Spain
// Rev: 1, mar 2011

//`include "u_control.v"
//`include "u_datos.v"

module multiplicador(
 input  [3:0] a,      // Entrada del dato A
 input  [3:0] b,      // Entrada del dato B
 input  clk,          // Reloj del sistema
 input  xs,           // Señal de comienzo
 input  reset,        // Reset del sistema
 output fin,          // Señal de fin de operacion
 output [7:0] databus // Bus de datos de resultado (parcial y final)

 );

// Conexiones internas para interconectar los dos módulos
wire wa,wd,upcont,wsuml,wsumh,cycont,clsincr,shrsum;


// Instancia de la unidad de datos, debe completar las conexiones
u_datos  U_DATOS(
 .data(a)
 
        // Complete aquí todas las conexiones que faltan
        // Para ello puede abrir el fichero incluido: "u_datos.v"
        // y mirar la definición del módulo "u_datos" que sirve 
        // como ejemplo de interconexión de módulos.
  );

/* Aqui debe instanciar la unidad de control del mismo
   modo que se ha instanciado la unidad de datos.
   Debe interconectar todas las conexiones 
   con la unidad de datos y las entradas y salidas. */



endmodule



// Design: Test bench - Multiplicador secuencial
// Description: Testbench para algunos casos
// Author: Paulino Ruiz de Clavijo Vázquez <paulino@dte.us.es>
// Copyright Universidad de Sevilla, Spain
// Rev: 1, feb 2011

`timescale 1ns / 1ps
// `include"multiplicador.v"

module test();

 reg [3:0] a; // Numero a
 reg [3:0] b; // Numero b
 reg xs;
 reg clk=0;
 reg reset=0;
 wire fin;
 wire [7:0] databus;

 reg [8:0] n;
 reg [3:0] a_vec[0:9]; // Vector de 10 elementos de prueba
 reg [3:0] b_vec[0:9]; // Vector de 10 elementos de prueba

 // Multiplicador
 
 multiplicador U_MUL(
   .a(a),.b(b),.clk(clk),
   .xs(xs),.reset(reset),
   .fin(fin),.databus(databus)
   );

 // Reloj 50 Mhz
 always
   #10 clk = ~clk;

 // Contenido de la memoria de test

 initial 
   begin
    $dumpfile("test.vcd"); // Esto permite usar Icarus
    $dumpvars(0, test);
    
    /* Inicialización del array de valores del test
       se puede leer desde un fichero de texto, pero
       por no complicarlo más prefiero hacerlo asi
    */
    {a_vec[0],b_vec[0]}=8'h00; // 0*0
    {a_vec[1],b_vec[1]}=8'h22; // 2*2
    {a_vec[2],b_vec[2]}=8'h57; // 5*7
    {a_vec[3],b_vec[3]}=8'hAA; // 10*10
    {a_vec[4],b_vec[4]}=8'hFF; // 16*16
    {a_vec[5],b_vec[5]}=8'hA0; // 10*0
    {a_vec[6],b_vec[6]}=8'h0A; // 0*10
    {a_vec[7],b_vec[7]}=8'hB1; // 11*1  
    {a_vec[8],b_vec[8]}=8'h1B; // 1*11
    {a_vec[9],b_vec[9]}=8'h2D; // 2*13


    reset=1;
    xs=0;
    #25 reset=0;
    // Salida para gtkwaves, se puede simular con Icarus o con Xilinx
    for(n=0;n<10;n=n+1)
     begin
       a=a_vec[n];
       b=b_vec[n];
       @(negedge clk);
       xs=1; 
       @(negedge clk);
       xs=0;

       /* Lo correcto es esperar la señal de fin con

          @(negedge fin); 

          pero si la unidad está mal este test se bloquea
          por eso es preferible esperar 500ns
       */
       #500;
       if((databus)!=a*b)
        $display("Error en el resultado de la multiplicacion %x x %x",a,b);
      end
     $finish;
   end
endmodule






`timescale 1ns / 1ps

// Design: Multiplicador secuencial

// Description: Preifericos, y modulo top del diseño

// Author: Enrique Ostua <ostua@dte.us.es>, Paulino Ruiz de Clavijo Vázquez <paulino@dte.us.es>

// Copyright Universidad de Sevilla, Spain

// Rev: 1, mar 2011



/*

       1             Los segmentos son activos en bajo

      ---

   6 |   | 2

      ---   <- 7

   5 |   | 3

      ---

       4

*/



module sseg(

  input wire [3:0] d,

  output reg [7:1] seg

  );



  always @*

  begin

  case (d)

      4'h0: seg = 7'b1000000; //--0

      4'h1: seg = 7'b1111001; //--1

      4'h2: seg = 7'b0100100; //--2

      4'h3: seg = 7'b0110000; //--3

      4'h4: seg = 7'b0011001; //--4

      4'h5: seg = 7'b0010010; //--5

      4'h6: seg = 7'b0000010; //--6

      4'h7: seg = 7'b1111000; //--7

      4'h8: seg = 7'b0000000; //--8

      4'h9: seg = 7'b0010000; //--9

      4'hA: seg = 7'b0001000; //--A

      4'hB: seg = 7'b0000011; //--b

      4'hC: seg = 7'b1000110; //--C

      4'hD: seg = 7'b0100001; //--d

      4'hE: seg = 7'b0000110; //--E

      default: seg = 7'b0001110; //--F

    endcase

  end

endmodule  // sseg



// Modulo de refresco del display de 4 digitos 

// El reloj hay que conectarlo al de la placa

module display_controller(

  input  wire clk,               // Reloj para el refresco

  input  wire [3:0] d3,d2,d1,d0, // 4 digitos de entrada

  output wire [7:1] seg,         // Segmentos de salida

  output wire dp,                // Punto del segmentos

  output reg [3:0] an            // Anodos de los segmentos

   );



  reg [15:0] displcounter=0;   // Divisor de frecuencia

  reg [3:0] d;                 // Entrada BCD 

  

  assign dp = 1'b0; // Punto decimal: 0 ON, 1 OFF



  always @(posedge clk)

   begin

     displcounter = displcounter + 1;

   end

  

  always @(displcounter,d0,d1,d2,d3)

  begin

    case ({displcounter[15],displcounter[14]})

      2'b00:   begin d = d0; an = 4'b1110; end

      2'b01:   begin d = d1; an = 4'b1101; end

      2'b10:   begin d = d2; an = 4'b1011; end

      default:  begin d = d3; an = 4'b0111; end

    endcase

  end

  // Instancia del convertidor

   sseg U_bcdconverter(d, seg);

endmodule // display_controller





module sistema_completo(

  input wire clk,        // Reloj placa

  input wire clkext,     // Reloj boton

  input wire start,      // Generador de xs

  input wire reset,      // Reset de la unidad de control

  input wire [3:0] data, // Conectados directamente a los switches

  input wire [3:0] datb, // Conectados directamente a los switches

  output wire [6:0] seg, // Segmentos del displays

  output wire dp,        // Punto del display

  output wire [3:0] an,  // Anodos de los displays

  output reg[7:0] ld     // Leds

   );

  

  wire [7:0] busc; // Bus interno conectado a la unidad de datos

  reg xs=0;        // Señal para el comienzo

  wire fin;

  

  display_controller U_display(clk,

       data,  // Numero A conectado directamente desde los switches

       datb,  // Numero B conectado directamente desde los switches

       busc[7:4],busc[3:0], seg, dp, an);

  

    multiplicador U_multiplicador(

    .clk(clkext),

    .a(data),

    .b(datb),

    .xs(xs),

    .reset(reset),

    .fin(fin),

    .databus(busc));

    

  // Este proceso controla la señal xs a partir del pulsador

  always@(posedge start,posedge clkext)

  begin

     if(start)  

      xs <=1;   

     else

      xs <=0;

  end

  

  // Este proceso controla los leds.

  // Con la señal de fin se activan todos los leds

  always@(posedge clkext,posedge fin,posedge reset)

  begin

    if(reset)

      ld <= 8'b00000000;

    else if(fin)

      ld <= 8'b11111111;

    else if(xs)

      ld <= 8'b10000000;

    else

      ld <= {ld[0],ld[7:1]};  

  end

endmodule




// Design: Multiplicador secuencial

// Description: Unidad de control del multiplicador

// Author: Paulino Ruiz de Clavijo Vázquez <paulino@dte.us.es>

// Copyright Universidad de Sevilla, Spain

// Rev: 1, mar 2011



module u_control(

 input xs,clk,reset,busc0,cycont,

 output reg clsincr,fin,

 output reg wa,wd,wsuml,wsumh,shrsum,upcont

    );





 // Defina aqui la lista de estados con la sentencia parameter

  

 ...



 // Defina aqui dos variables  tipo reg llamadas: 

 //     estado_actual y siguiente_estado

 // Dimensione ambas variables con el tamaño correcto en función

 // del número de estados que definió previamente



   

 ...



 // Proceso de cambio de estado, utiliza la variables

 // definidas previamente

 always @(posedge clk,posedge reset)

  begin

   if(reset)

   estado_actual <= S0;

  else

   estado_actual <= siguiente_estado;

  end

  

  // Proceso combinacional de calculo de proximo estado

  // debe obtenerlos a partir de la carta ASM 

  always @(*)

  begin

   clsincr=0; // Por defecto se establecen todas las señales a cero

   wa=0;

   wd=0;

   wsuml=0;

   wsumh=0;

   upcont=0;

   shrsum=0;

   fin=0;

   case(estado_actual)

         // Rellene todos los estados y active las señales de control

         // siguiendo la carta ASM

      S0:

        if(xs)

          begin

          clsincr=1;

          siguiente_estado = S1;

          end

        else

          siguiente_estado = S0;



       // Aqui deben ir el resto de estados

       ....

       ///



      default:  // No elimine esta sentencia para evitar problemas

          siguiente_estado=S0;  // Cuidado aqui, hay que cubrir todos los casos 

                                // de lo contrario aparece un latch

  endcase

  end

endmodule




// Design: Multiplicador secuencial

// Description: Multiplicador secuencial - Unidad de datos

// Author: Paulino Ruiz de Clavijo Vázquez <paulino@dte.us.es>

// Copyright Universidad de Sevilla, Spain

// Rev: 1, feb 2011



// Registro A

module reg_a (

 input clk,cl,w,

 input [3:0] din,

 output reg [3:0] dout

 );

 

 always@(posedge clk)

  begin

    if(cl)

    dout <= 0;

   else if(w)

    dout <= din;

  end

endmodule



// Sumador de 4 bits

module sumador_4(

 input [3:0] a,

 input [3:0] b,

 output [3:0] res,

 output cout

 );

 assign {cout,res} = a + b;

endmodule



// Contador modulo 4

module cont_4(

 input clk,cl,up,

 output cy

 );

 

 reg [1:0] q;

 

 always@(posedge clk)

 begin

  if(cl)

    q <= 0;

  else if(up)

    q <= q + 1;

 end

 assign cy = q[0]&q[1];

endmodule



// Registro de desplazamiento

module reg_despl(

 input clk,cl,shr,w,sri,

 input [3:0] din,

 output reg [3:0] dout);

 

 always@(posedge clk)

 begin

  if(cl)

   dout <= 0;

  else if(w)

   dout <= din;

  else if(shr)

   dout <= {sri,dout[3:1]};

 end

endmodule



// Biestable D para capturar el carry del sumador

module biestable_d(

 input clk,cl,d,w,

 output reg q);

 

 always @(posedge clk)

 begin

  if(cl)

   q <= 0;

  else if(w)

   q <= d;

  else

   q <= q;

 end

 

endmodule





/* Unidad de datos - Descripcion estructural

 * Se instancian todos los componentes de la unidad

 * de datos y se interconectan entre si

 */

module u_datos(

 input [3:0] data, // Dato a

 input [3:0] datb, // Dato b

 input clk,wa,upcont,clsincr,

 input wd,wsumh,wsuml,shrsum,

 output [7:0] busc, // Bus de salida

 output cycont

);



 wire [3:0] busa;

 wire [3:0] busb;

 wire cable_and,q,cout;



 // Instancia del registro A

 reg_a A(.din(data),.cl(clsincr),

         .clk(clk),.w(wa),.dout(busa));

 // Instancia del sumador

 sumador_4 SUMADOR(.a(busa),.b(busc[7:4]),

                   .res(busb),.cout(cout));

 // Instancia del registro de desplazamiento para SUMH             

 reg_despl SUMH(.din(busb),.cl(clsincr),

                .w(wsumh),.clk(clk),.shr(shrsum),

                .sri(q),.dout(busc[7:4]));

 // Otra instancia del registro de desplazamiento para SUML

 reg_despl SUML(.din(datb),.cl(clsincr),

                .w(wsuml),.clk(clk),.shr(shrsum),

                .sri(busc[4]),.dout(busc[3:0]));

 // Instancia del biestable d           

 biestable_d BIESTABLE_D(.clk(clk),.cl(clsincr),.w(wd),

                       .d(cable_and),.q(q));

 // Instancia del contador

 cont_4 CONT_4(.clk(clk),.cl(clsincr),.cy(cycont),

               .up(upcont));

 // Una simple puerta and de 2 entradas               

 and P_AND(cable_and,busc[0],cout);



endmodule





